Nim game

You are encouraged to solve this task according to the task description, using any language you may know.
Nim is a simple game where the second player ─── if they know the trick ─── will always win.



The game has only 3 rules:

   start with   12   tokens
   each player takes   1,  2,  or  3   tokens in turn
  the player who takes the last token wins.


To win every time,   the second player simply takes 4 minus the number the first player took.   So if the first player takes 1,   the second takes 3 ─── if the first player takes 2,   the second should take 2 ─── and if the first player takes 3,   the second player will take 1.



Task

Design a simple Nim game where the human player goes first, and the computer always wins. The game should enforce the rules.




tokens = 12
 
print "Nim Game"
print "Starting with " + tokens + " tokens."
print
 
printRemaining = function()
	print tokens + " tokens remaining."
	print
end function
 
playerTurn = function(take)
	take = floor(val(take))
	if take < 1 or take > 3 then
		print "Take must be between 1 and 3."
		return false
	end if
 
	globals.tokens = tokens - take
 
	print "Player takes " + take + " tokens."
	printRemaining
	return true
end function
 
computerTurn = function()
	take = tokens % 4
	globals.tokens = tokens - take
 
	print "Computer takes " + take + " tokens."
	printRemaining
end function
 
while tokens > 0
	if playerTurn(input("How many tokens would you like to take? ")) then
		computerTurn
	end if
end while
 
print "Computer wins."